

# GNUPLOT : 区関数グラフ

## 1 関数グラフの部分表示

例として、関数  $y = \frac{1}{2}(x+1)^2$  を  $-1 \leq x \leq 0$  の範囲で描画する場合を考えよう。

プログラム

```
fn.plt
reset
set terminal pdfcairo enhanced
set output "fn.pdf"

unset key
set size ratio -1
set xzeroaxis
set yzeroaxis
set grid
set xrange [-1.5:2.5]
set yrange [0:1]
plot sample [-1:0] (x+1)**2/2 lt 1 lw 3
```

結果




図 1: 関数の部分表示

```
plot sample [-1:0] (x+1)**2/2 lt 1 lw 3
```

の sample を取って、

```
plot [-1:0] (x+1)**2/2 lt 1 lw 3
```

とすると、 $x$  軸の範囲は、 $[-1.5:2.5]$  よりも  $[-1:0]$  が優先されて、以下のようなになる。




図 2: fn.plt の plot から sample を取り除いた場合

## 2 区分ごとに定義された関数

次に、区分ごとに定義された関数、例えば、

$$f(x) = \begin{cases} \frac{(x+1)^2}{2} & -1 \leq x \leq 0 \\ -x^2 + x + \frac{1}{2} & 0 \leq x \leq 1 \\ \frac{(x-2)^2}{2} & 1 \leq x \leq 2 \end{cases} \quad (1)$$

の表示について考える。

この場合には、plot 部分を、

```
plot sample [-1:0] (x+1)**2/2 lt 1 lw 3, [0:1] -x**2+x+0.5 lt 1 lw 3, \
[1:2] (x-2)**2/2 lt 1 lw 3
```

と書けば良い。

結果は、




図 3: (1) 式で定義された関数

さらに、 $x \leq -1$  と  $x \geq 2$  では、 $f(x) = 0$  と明示的に表したいときは、plot に

```
plot sample [*:-1] 0 lt 1 lw 3, [-1:0] (x+1)**2/2 lt 1 lw 3, \  
[0:1] -x**2+x+0.5 lt 1 lw 3, [1:2] (x-2)**2/2 lt 1 lw 3, [2:*] 0 lt 1 lw 3
```

と、`[*:-1] 0 lt 1 lw 3` と `[2:*] 0 lt 1 lw 3` を付け加えれば良い。  
結果は、以下の通りである。




図 4: (1) 式の定義域以外を 0 と置いた場合

### 3 区分関数の定義：三項演算子

(1) 式の関数  $f(x)$  を定義する。

```
f(x) = (x>-1 && x<0) ? (x+1)**2/2 : (x<1) ? -x**2+x+0.5 : \  
(x<2) ? (x-2)**2/2 : 1/0
```

$x > 2$  は、関数の定義外となるので、何もプロットしない。その命令が `1/0` である。  
改めてプログラムの全体を表示すると、以下ようになる。

```
fn3.plt  
  
reset  
set terminal pdfcairo enhanced  
set output "fn3.pdf"  
  
unset key  
set size ratio -1  
set xzeroaxis  
set yzeroaxis  
set grid  
set xrange [-1.5 : 2.5]  
set yrange [0:1]  
f(x) = (x>-1 && x<0) ? (x+1)**2/2 : (x<1) ? -x**2+x+0.5 : \  
(x<2) ? (x-2)**2/2 : 1/0  
plot f(x) lt 1 lw 3
```

アウトプットは、もちろん n、図 (3) と同じである。

一方、図(4)を描くためには、関数  $f(x)$  の定義を、

$$f(x) = (x < -1) ? 0 : (x < 0) ? (x+1)**2/2 : (x < 1) ? -x**2+x+0.5 : \backslash \\ (x < 2) ? (x-2)**2/2 : 0$$

と変更する。